



Bulletin Technique

Janvier Février Mars 1990

La revue du support technique d'Aware



MOBILE COMPUTER

- Formats de fichiers



ORGANISEUR II

- Logiciel d'inventaire
- Protéger un Datapak
- Transfert par Modem
- Modification du menu principal
- Organiseur ou Organiseur ?
- Aide à la Programmation



PC4

- La commande ' isole ' d'Archive
- Feuilles de styles Quill

Edité par : Aware
7/9, rue des Petites Ecuries
75010 PARIS
Tél. : (1) 45.23.21.12
Télex : 281 941
Fax : (1) 45.23.02.37
3616 code Aware



Mieux vaut tard que jamais !

Vous ne devez pas en croire vos yeux ! Pourtant c'est vrai, vous avez enfin le dernier Bulletin Technique Psion entre les mains. Je sais cependant qu'à l'agréable surprise s'ajoute une petite déception quand vous lisez "Janvier Février Mars 1990" alors que nous sommes en... juillet !


Justement, nous profitons de ce début d'année (!) pour changer l'aspect de notre revue et avec cette nouvelle maquette, nous avons également pris certaines résolutions : nous allons rattraper notre retard très rapidement (ne vous étonnez donc pas de recevoir plusieurs bulletins rapprochés) et surtout, nous essayerons de ne plus du tout être en retard !

En ce qui concerne la nouvelle maquette, seule la présentation change réellement car vous continuerez à trouver dans notre revue les trucs et astuces vous permettant d'exploiter au mieux nos produits, des exemples de programmes en OPL quel que soit votre niveau de programmation, des informations techniques variées répondant à vos questions les plus fréquentes, etc.


Et n'oubliez pas que c'est en restant à l'écoute de vos besoins et de vos suggestions que nous réalisons ces articles et essayons constamment de les améliorer. Alors, n'hésitez pas à nous faire part de toutes vos remarques sur le Minitel 3616 code Aware.

A très bientôt.

Martine Carrigues



Les formats de fichiers du MC



2

Le MC utilise un certain nombre de formats de fichiers pour stocker ses données. On les retrouve avec l'Agenda, la Base de données, l'OPL et l'Editeur de textes. Ce sont les fichiers Agenda et Base de données que nous allons examiner ci-après.

Le MC, lors de la création d'un fichier, enregistre sa structure. Pour permettre l'ouverture de celui-ci avec l'OPL, il faut, lors de l'ouverture, lui réindiquer la structure du fichier. La base de données est constituée de 32 champs de données alphanumériques. Sa déclaration d'ouverture est la suivante :

Proc test :

```
proc test:
OPEN
"M:\BDD.DBF",A,C1$,C2$,C3$,C4$,C5$,C6$,
C7$,C8$,C9$,C10$,C11$,C12$,C13$,C14$,C15$,
C16$,C17$,C18$,C19$,C20$,C21$,C22$,C23$,C24$,
C25$,C26$,C27$,C28$,C29$,C30$,C31$,C32$
do
    print a.c1$
next
until eof
endp
```

Ce petit programme ouvre le fichier standard de la Base de données pour le lire.

Le fichier Agenda

Lui, a un nombre moins grand de champs avec en plus, des données de types numériques. Voici sa structure :

Entier	Jour
Entier	Heure de rendez-vous
Entier	Durée du Rendez-vous
Entier	Présence d'une alarme ou d'un message vocal
Entier	Heure de l'alarme, le cas échéant
Chaine	Texte de l'alarme

Les jours commencent au 1/1/1900 (jour 0). L'Agenda ne gère pas les dates avant le 5 janvier 1970.

Si l'entier représentant les heures est positif, il s'agit en fait d'une note et l'entier est le numéro d'ordre de la note.

La durée du rendez-vous n'est pas significative s'il s'agit d'une note.

Si l'entier indiquant une alarme est égal à 1, il s'agit d'une alarme sonore.

S'il est égal à quatre, il s'agit d'un message vocal.

L'heure de l'alarme doit être inférieure ou égale à l'heure de rendez-vous.

S'il n'y a pas d'alarme, l'entier n'est pas significatif. Le texte de l'alarme ne doit pas faire plus de 64 caractères.

Voici, en exemple, un programme lisant le fichier agenda et affichant les différents rendez-vous dans l'ordre de saisie :

Proc Agenda :

```
Proc Agenda:
global a%,he%
screen 70,10
open "m:\agenda.dry",a,a%,b%,c%,d%,e%,f$
do
    at 1,1
    print "Le",genjo$:(a.a%)
    if a.b%>=0
        print "Notes no",a.b%
    else
        he%=32768+a.b%
        print "Entrée de l'agenda à", :genhe$:(he%)
        print "d'une durée de", :genhe$:(a.c%)
    endif
    if ((a.e% and 1) and not (a.e% and 2))
        print "Alarme sonore à", :genhe$:(a.d%)
    endif
    if (a.e% and 4)
        print "Alarme vocale à", :genhe$:(a.d%)
    endif
    print
```

3

Suite du programme

```

    print "Texte :",a.f$;rept$(" ",64)
    a%=get
    next
until eof
endp


proc genhe$(a%)
    rem conversion d'un entier en Heures Minutes et affichage
    rem heure 0 : 0:00

    print int(a%/60),"heures",
    print ((a%/60.)-int(a%/60))*60,"minutes"
endp

proc genjo$(jour%)
    rem conversion d'un entier en date et affichage
    rem date 0 : 1/1/1900

    local a%
    global année%,jours%,mois%
    a% = jours%
    if jours%<0
        a%=jours%+65537
    else
        a%=jours%+1
    endif
    année%=int(a%/365.25)
    a%=a%-int(365.25*année%)
    année%=année%+1900
    if a%>334-(année%/4.=année%/4)
        jours%=a%-334+(année%/4.=année%/4) : mois%=12
    elseif a%>303-(année%/4.=année%/4)
        jours%=a%-303+(année%/4.=année%/4) : mois%=11
    elseif a%>273-(année%/4.=année%/4)
        jours%=a%-273+(année%/4.=année%/4) : mois%=10
    elseif a%>243-(année%/4.=année%/4)
        jours%=a%-243+(année%/4.=année%/4) : mois%=9
    elseif a%>212-(année%/4.=année%/4)
        jours%=a%-212+(année%/4.=année%/4) : mois%=8
    elseif a%>181-(année%/4.=année%/4)
        jours%=a%-181+(année%/4.=année%/4) : mois%=7
    elseif a%>151-(année%/4.=année%/4)
        jours%=a%-151+(année%/4.=année%/4) : mois%=6
    elseif a%>120-(année%/4.=année%/4)
        jours%=a%-120+(année%/4.=année%/4) : mois%=5
    elseif a%>90-(année%/4.=année%/4)
        jours%=a%-90+(année%/4.=année%/4) : mois%=4
    elseif a%>59-(année%/4.=année%/4)
        jours%=a%-59+(année%/4.=année%/4) : mois%=3
    elseif a%>31
        jours%=a%-31 : mois%=2
    elseif a%>0
        jours%=a% : mois%=1
    endif
    print dayname$(dow(jours%,mois%,année%)),
    print jours%;"/";mois%;"/";année%
endp

```



Par
Gilles JEAN

Un logiciel d'inventaire...

C'est à cor et à cri que vous nous réclamez un logiciel d'inventaire pour votre Organiseur II. Le logiciel, dont vous trouverez les sources ci-dessous, a été mis au point pour vous permettre de passer en revue rapidement votre stock. Dans ce souci, nous nous permettons de ne pas importer de fichier de référence pour ne pas alourdir notre tâche par des temps d'accès en recherche plus ou moins longs. La vérification de la présence des codes articles lus étant sous-traitée à un micro-ordinateur PC, MAC ou autre par l'intermédiaire d'un logiciel particulier après avoir importé notre fichier de l'Organiseur II vers un micro-ordinateur.

Invent :

```

Invent:
local b$(255),q%,k%
escape off
if nobar:
intro:
file:
clock:
at 1,2
print "Passez le lecteur..."
last
b$=a.code$
do
kstat 3
at 1,3

```

Suite du programme

```
print chr$(22);left$(b$,20)
if count
at 1,4
print chr$(23);pos;"/";count,"Codes lus"
endif
b$=bar$(511,-1)
k%=key
if k%=0
beep 99,99
a.code$=right$(b$,len(b$)-1)
append
elseif k%=2 or k%=1
break
elseif k%=3 or k%=5 and pos>1
back
elseif k%=4 or k%=6 and count>1 and pos<count
next
elseif k%=8 and count
beep 50,50
at 1,4
print chr$(23);"Confirmer? O/N"
if get=%4
erase
ifeof
last
endif
endif
at 1,4
print chr$(23)
elseif k%=13
at 1,3
print chr$(22);"Code?",
trap input b$
if err=0 and len(b$)
a.code$=b$
append
endif
endif
if space<5000
beep 200,800
beep 300,300
beep 200,800
view(4,"PLUS DE MEMOIRE. Veuillez effectuer le transfert -MERCI.")
break
b$=a.code$
until 0
endif
```

La procédure NOBAR listée ci-dessous permet de vérifier la présence du lecteur de codes à barres sur votre Organiseur II. Si cela n'est pas le cas, un message est affiché et le programme retourne au menu d'appel.

```
nobar:
onerr bar::
bar$:(1,1)
```

Suite du programme

```
return 1
bar::
cls
print"Lect. Codes à Barres";
print" Non disponible."
print chr$(25);
print"Tapez sur une touche"
get
return
```

Cette routine permet un affichage de présentation ou d'accès.


```
intro:
local b%
at 1,1
print rept$("*",20)
do
b%=b%+1
at 1,2
print left$("**** INVENTAIRE ****",b%)
beep 99,99
pause 10
until b%=20
at 1,4
print" Bulletin technique"
pause 50
```

Cette routine permet de créer, s'il n'existe pas, ou d'ouvrir le fichier INVENT qui contient les codes à barres lus.


```
file:
trap create"a:invent",a,code$
trap open"a:invent",a,code$
```

```
clock:
cls
print rept$(chr$(2),15)
clock(1)
```

L'utilisation de ce produit est très simple et dans le plus pur esprit de l'Organiseur II. Une fois le logiciel lancé (INVENT), il ne vous reste plus qu'à user et abuser de votre lecteur afin de stocker l'inventaire. Néanmoins, vous pouvez visualiser les lectures déjà effectuées avec les touches curseurs, détruire une lecture avec la touche DEL (avec confirmation) et saisir manuellement un code dont l'étiquette ne serait pas visible avec la touche EXE. La sortie de l'inventaire se fait par la touche ON. Afin de vous permettre d'effectuer le transfert du fichier des codes lus, nous réservons 5000 caractères (pour la Comms-Link) et lorsque cette limite est atteinte, un message apparaît et la saisie est interrompue avec retour au menu précédent. Maintenant à vous de jouer...



Protéger un Datapak contre la copie et l'écriture



Par
Alexandre BOUILLOT

Il existe plusieurs façons de protéger un Datapak contre l'écriture, la copie ou les deux :

1/ Utiliser un Copieur 8 pour copier le Datapak et sélectionner les options de protection. Cette technique n'est accessible qu'aux possesseurs d'un Copieur 8.

2/ Réaliser, à l'aide du développeur sur PC, une image du Datapak avec les options de protection. Cette technique suppose la création d'une image de Datapak et de sa copie sur un Datapak formaté de l'Organiseur.


3/ Protéger un Datapak présent dans l'Organiseur contenant déjà des données. Cette méthode ne nécessite pas d'équipement supplémentaire en matériel ou en logiciel en plus de l'Organiseur et du Datapak. Elle est particulièrement valable lorsque l'on ne peut accéder à un Copieur 8 ou à un Développeur. C'est cette méthode que nous allons étudier ici avec un petit programme en langage machine pour écrire les bits nécessaires sur le Datapak.

Attention :


Lors de la recopie de ce programme, une erreur dans la saisie des codes peut entraîner un "plantage" irrémédiable de la machine. Lorsque le Datapak est protégé, il n'est plus possible de le déprotéger. Il faudra alors le reformatter dans un effaceur UV. Ce programme n'est valable que pour les Datapak. Si vous l'utilisez avec un Rampak, celui-ci sera reformatté.

CPWRT :

```
CPWRT :
LOCAL IN%,D%,A%(23)
A%(1)=$3F62
A%(2)=$2521
A%(3)=$CE00
A%(4)=$005F
A%(5)=$3F60
A%(6)=$3F5D
A%(7)=$C4FF
A%(8)=$F721
A%(9)=$88CE
A%(10)=$0000
A%(11)=$5F3F
A%(12)=$60CE
A%(13)=$2188
A%(14)=$CC00
A%(15)=$013F
A%(16)=$6125
A%(17)=$08CE
A%(18)=$0000
A%(19)=$39CE
A%(20)=$00F6
A%(21)=$39CE
A%(22)=$00C0
A%(23)=$3900
IN%=MENU("COP, ECR, COPECR, FIN")
IF IN%=1
A%(7)=$C4DF
ELSEIF IN%=2
A%(7)=$C4F7
ELSEIF IN%=3
A%(7)=$C4D7
ELSE
RETURN
ENDIF
PRINT " PLACER LE PACK"
PRINT " EN B:"
PAUSE 40
CLS
PRINT "APPUYEZ SUR EXE"
PRINT " POUR LANCER"
IN%=GET
IF IN%=13
D%=USR(ADDR(A%()),1)
CLS
IF D%
PRINT ERR$(D%)
ELSE
CLS
PRINT " PACK B:"
PRINT " PROTEGE"
ENDIF
GET
ENDIF
RETURN
```



Transfert de données par modem



Par
Gilles JEAN


Véritable phénomène de société, l'Organiseur II n'en finit pas de faire parler de lui, et si ce fait est largement constatable, il n'en demeure pas moins qu'il possède de très nombreux arguments pour communiquer (jeu de mots).

Si cette introduction vous paraît confuse, la suite de cet article va essayer de vous renseigner le plus précisément possible et avec le plus de clarté possible sur la puissance de communication de l'Organiseur II. Cette machine a été conçue pour emporter avec soi un certain nombre d'informations qui peuvent avoir été saisies sur son clavier, qui peuvent provenir d'un autre Organiseur II (Datapak ou Rampak) ou bien qui peuvent avoir été saisies sur un Ordinateur (par le biais d'un traitement de texte, d'un tableur, d'une base de données, etc...) puis transférées par la liaison RS232 Comms-Link. Voilà, le mot est lâché.

Le principe de la communication RS232 est très simple. Un poste émetteur envoie à un poste récepteur des éléments d'information constituant par exemple un fichier. Pour que cela fonctionne correctement, les deux postes doivent se mettre d'accord sur la vitesse de communication, la parité, le nombre de bit de données effectives pour un caractère, le nombre de bits indiquant la fin d'un train de bits correspondant à un caractère, la même technique de contrôle de débit et éventuellement le même protocole de communication (méthode de validité des données). Pour que tout ce que nous venons d'énumérer soit possible, il faut que chacune des machines soit équipée d'une interface RS232 et d'un logiciel de communication.

La communication entre un PC/MAC et un Organiseur II en direct (sans modem) a été largement traitée dans les précédents Bulletins Techniques et dans la documentation de la liaison RS232 Comms-Link de l'Organiseur II. Nous ne reviendrons pas dessus.

La communication à la norme RS232 sous entend qu'un élément DTE (Data Terminal Equipment) ne communique qu'avec un DCE (Data Communicating Equipment). Un PC ou un Mac seront donc un DTE et un modem sera un DCE. La plupart des utilisateurs se servent de l'Organiseur en connexion directe avec un ordinateur de bureau (PC, Mac, etc...). Il était donc plus simple de le faire considérer comme un DCE plutôt que comme un DTE de par le simple fait que cela évite de rajouter un câble nul-modem entre les deux équipements. Le connecteur de la DB25 femelle de Comms-Link se présente donc comme un connecteur de DCE. Pour transférer un fichier d'un Organiseur II à un PC et inversement via le réseau commuté des P&T, vous devez doter vos deux machines d'un modem. Les lignes téléphoniques sont particulièrement mal adaptées aux transferts de signaux numériques. Un modem module les signaux numériques à émettre pour les transformer en signaux analogiques lorsqu'il émet et inversement lorsqu'il reçoit. La plupart des modems actuellement sur le marché utilise le jeu de commandes mis au point par la société HAYES. Ces commandes sont utilisées par de nombreux logiciels de communication et figurent dans les notices des modems. Il existe aujourd'hui un certain nombre de modems disposant chacun de fonctionnalités et de caractéristiques qu'il est important de connaître avant de tenter votre premier transfert.




Pour connecter votre Organiseur II à un modem, vous devez ajouter à votre Comms-link un adaptateur (nul-modem) qui intervertit trois paires de lignes de signaux :

TD avec RD
RTS avec CTS
DTR avec DSR

Si l'Organiseur II est à l'origine de l'appel, la liaison entre les broches DTR et DSR empêchera le modem de fonctionner. Cette règle n'est pas valable dans tous les cas. Certains modems maintiennent d'eux-mêmes leur propre signal DTR et dans ce cas, pas de problème. Mais la quasi-totalité des modems pockets (portables, et c'est bien là l'idéal pour l'Organiseur II) ne maintiennent pas ce signal dans un but évident d'économie d'énergie. Il faut donc doter votre équipement d'un adaptateur modem pour Organiseur II qui aura pour fonction de maintenir le signal DSR haut afin de ne pas couper la communication. Et ainsi, n'importe quel Organiseur II pourra communiquer avec un PC, un Mac ou un autre périphérique doté d'une interface RS232 par le réseau commuté des P&T. Vous noterez que nous ne pouvons divulguer le schéma de connexion de cet adaptateur dans la mesure où celui-ci est régi par des droits d'auteur et ainsi légalement protégé.

Imaginons, par exemple, que vous disposiez d'un PC avec une ligne téléphonique directe (pourquoi pas un numéro vert) qui permette à votre équipe commerciale de vous communiquer, avec un Organiseur II, un modem et une ligne téléphonique, le fichier des commandes prises dans la journée. Ces commandes sont alors expédiées dans la soirée et vos clients sont livrés dès le lendemain matin. Intéressant non !!!



Il suffit alors de connecter un modem disposant du jeu d'instructions HAYES sur votre PC et de lancer le logiciel de communication CL livré avec chaque câble Comms-Link.

Exemple : CL 2400,1,2,ATZ\ATS0=3\


Cela permet de lancer le logiciel de communication à une vitesse de 2400 bauds, sur le port COM1, en attente de porteuse de données (DCD), en réinitialisant le modem et en le configurant en réponse automatique après 3 sonneries (consulter le manuel Comms-Link et le fichier readme.doc de la disquette pour de plus amples renseignements). Le programme côté Organiseur II pourrait donc être le suivant :

Comm :


```
comm:
LOCAL a$(254),a%(2)
a%(1)=$3f17
a%(2)=$3900
usr(addr(a%()),0)
LSET:(2400,0,8,1,0,-1,-
1,chr$(13)+chr$(10),chr$(26),"",chr$(13),chr$(26-
),"",10,2)
ONERR jnk::
WHILE 1
JNK::
a$=LINPUT$(254,1)
IF a$="" :BREAK :ENDIF
ENDWH
ONERR OFF
a$=TRIG$(254,10,"ATZ"+CHR$(13))
a$=LINPUT$(254,10)
IF a$="OK"+CHR$(13)+CHR$(10)
a$=TRIG$(254,10,"ATDP45232112"+CHR$(13))
a$=LINPUT$(254,120)
a$=LINPUT$(254,10)
rem appel automatique d'un numero de telephone
rem ici, c'est le numero du standard AWARE
rem de grace pour notre charmante standardiste
rem n'oubliez pas de changer ce numero avant vos essais
IF loc(upper$(a$),"CONNECT")
XTSEND:("COMM","COMM",0)
beep 99,99
PRINT "transfert ok"
else
PRINT "Impossible de"
PRINT "trouver le modem"
ENDIF
GET
ENDIF
```

Comme vous pouvez le constater, la fabuleuse machine que nous nommons Organiseur II a plus d'un tour dans son (petit) sac et sachez que le monde professionnel fait aujourd'hui bonne place à cette merveille de puissance et de miniaturisation.

Où se procurer les adaptateurs modem/Organiseur II ?
Printec - M. Rembliez (16 (1) 39 68 92 19 ou 16 (1) 30 36 82 33)
Agata 31 - M. Galicier (16 61 21 01 85)



Donner un "look" de LZ à vos XP ou CM




Par
Alexandre BOUILLLOT


Une des particularités des modèles LZ est de présenter des menus dont les options ont leur première lettre en majuscule et les autres en minuscules. Dans les menus des XP et CM, toutes les options sont en majuscules. Voici un petit programme qui permettra de transformer l'aspect des noms des options du menu principal.

LZ :

```
LZ:
LOCAL  START%, LONG%, A%
START%=PEEKW($2002)
WHILE PEEKB(START%)>0
LONG%=PEEKB(START%)
A%=START%+2
DO
POKEB A%,ASC(LOWER$(CHR$(PEEKB(A%))))
REM
REM A REMPLACER EN :
REM POKEB A%,ASC(UPPER$(CHR$(PEEKB(A%))))
REM POUR REVENIR A UN MENU NORMAL
REM
A%=A%+1
UNTIL A%>START%+LONG%
START%=A%+2
ENDWH
```



Par
Gilles JEAN



Organiseur ou Organiseur ?

Cette petite routine permet de savoir immédiatement le type de machine et le numéro de version du système d'exploitation de votre Organiseur II. Ce programme doit obligatoirement être transcodé en mode deux lignes (Xtran avec un LZ ou LZ64). En effet, en plus des informations affichées, cette routine permet de forcer le mode quatre lignes lorsqu'elle en a la possibilité. Voici donc le moyen d'écrire un seul programme pour les deux types de machines. Ami(e)s Développeurs, pensez au bonheur de n'avoir plus qu'une seule version pour vos programmes...


xplz :

```
xplz:
local f11%,f12%,mc%(2)
rem *****
rem Bulletin Technique
rem AWARE 1990
rem *****
if peekb($ffe8) and 8
f11%=1
if peekb($ffcb) and 128
f12%=1
endif
endif
if f11% and f12%
rem *****
rem adresses des routines systèmes
```


Suite du programme

```
rem a exécuter
rem *****
mc%(1)=$3f82
mc%(2)=$3900
usr(addr(mc%()),$100)
rem *****
rem la fonction USR permet de forcer le
rem passage au mode 4 lignes alors que
rem le programme a été compilé en mode
rem 2 lignes.
rem *****
rem *****
rem *****
rem Information sur les machines LZ
rem *****
print "Machine: LZ",
if peekb($ffe8)=13 and peekb($ffcb)=192
print "64"
else print
endif
rem *****
rem cette partie du programme permet
rem l'affichage du numéro de version
rem du système d'exploitation contenu
rem sur la rom de votre Organiseur II.
rem *****
print "Version:",
print left$(hex$(peekb($ffe9)),1);
print ".";
print right$(hex$(peekb($ffe9)),1)
else
rem *****
rem Information sur les machines CM/XP
rem *****
print "Machine:",
if peekb($ffe8)=0
print "CM"
else print "XP"
endif
rem *****
rem cette partie du programme permet
rem l'affichage du numéro de version
rem du système d'exploitation contenu
rem sur la rom de votre Organiseur II.
rem *****
print "Version:",
print left$(hex$(peekb($ffe9)),1);
print ".";
print right$(hex$(peekb($ffe9)),1)
endif
get
rem *****
rem fin du programme
rem Merci à l'équipe Psion pour l'aide
rem apporté à la réalisation de cette
rem routine.
rem *****
```

16



Par
Joël DROUIN



Aide à la programmation


Enfin un utilitaire qui va grandement vous aider dans la réalisation de vos programmes. Cet outil permet en effet, après avoir choisi son mode clavier, d'afficher simultanément la valeur ascii, la représentation en caractères et la correspondance hexadécimale du caractère tapé sur le clavier de votre Organiseur II.

Cod :


```
cod:
local k%
do
k%=menu("Alphamaj, Alphamin, Nummaj, Nummin")
if k%
cls
print "ASC CHR$ HEX$"
kstat k%
do
k%=get
at 2,1
print chr$(15);
print gen$(k%,-3)," ",chr$(k%)," ",hex$(k%)
until k%=1
endif
until k%=0
```

Il ne vous reste plus, maintenant, qu'à paufiner vos développements.

17



La commande 'isole' d'Archive



Par
Gilles JEAN

Il existe différents moyens pour retrouver un enregistrement dans un fichier Archive (cherche, trouve, suiv, pré). Ces commandes permettent d'accéder à un enregistrement de votre fichier de façon séquentielle. Cela signifie qu'Archive va partir de l'enregistrement en cours et va parcourir le fichier en chargeant enregistrement après enregistrement afin de vérifier si l'enregistrement en mémoire correspond à la recherche. Vous vous apercevrez rapidement que ces requêtes, effectuées sur un fichier de taille importante, prennent un certain temps.


La commande 'isole' vous permet de travailler sur une partie du fichier en cours. Vous pourriez donc travailler sur tous vos prospects résidant à Paris.

```
isole ville$='PARIS'
```


Cette commande va donc parcourir tout votre fichier afin de retenir les enregistrements pour lesquels la condition est vraie. Les enregistrements isolés sont classés selon la clef de tri en cours lors de la sélection. Cette commande va générer sur votre disque un fichier doté d'une extension 'IX5'. Après avoir isolé vos enregistrements, il vous faut activer votre fichier d'index.

```
active index 5
```

Vous noterez tout le bien fondé de cette commande qui lorsque le temps vous fait défaut, vous rendra d'immenses services. Archive est un produit offrant des fonctionnalités très riches. Consultez une fois encore votre manuel d'utilisation et vous vous rendrez compte que ce n'est pas sans raison si beaucoup de sociétés européennes s'en sont entièrement remises à lui.



Par
Gilles JEAN



Feuilles de styles

Ne cherchez pas. Il n'existe pas de fonction permettant de sauvegarder une feuille de style gérant le formatage de vos documents. Pourtant, la simplicité d'utilisation ainsi que la puissance de votre traitement de texte préféré vont vous permettre de remédier à ce problème.

Chacun des utilisateurs du traitement de texte Quill a, en effet, des besoins de formatages de textes, de tabulations, de format de papier, etc... qui sont différents de ceux des autres utilisateurs. La comptabilité utilisera tel type de format, le marketing tel autre...

Lorsqu'un utilisateur ouvre un nouveau document, Quill affecte automatiquement des valeurs par défaut pour un certain nombre de variables. Ainsi, le nombre de lignes par page est fixé automatiquement à 66 lignes, les tabulations sont placées de type gauche et tous les dix caractères, le bas de page est toujours centré avec l'affichage du numéro des pages en chiffres arabes, etc...

La solution consiste à avoir un document vierge contenant exclusivement les paramètres nécessaires à votre courrier. Le problème de mise en page en est donc considérablement réduit puisque tous les paramètres sont déjà correctement mis en place. Au revoir problème de marges, de tabulations, de formats, d'interlignage etc...

Comment mettre en place ce document ?
Nous allons créer un nouveau document que nous nommerons PERSO.

Lancer votre traitement de texte

c:\pc4>

Tapez alors quill puis validez. Un nouveau document est alors mis en mémoire et nous allons en modifier les paramètres.

Les marges :

F3
M
SPACE
5 fois flèche gauche
ENTER

Le format du document :

F3
F
H
70
ENTER
ENTER

Le bas de votre page :

F3
B
SPACE
1
ENTER
ENTER
ENTER
S

Vous avez sans doute déjà remarqué en consultant votre documentation qu'il existe beaucoup d'autres paramètres que nous aurions pu documenter. Il nous reste néanmoins encore une chose importante à effectuer : la sauvegarde de notre document.

F3
S
PERSO
ENTER
ENTER

Pour conclure, nous devons vous signaler qu'il existe une autre possibilité légèrement plus complexe à mettre en oeuvre. Le TSL permet effectivement d'enregistrer sur disque, donc dans un fichier, une suite de frappes au clavier. Ce qui nous amène à vous proposer un petit exercice : Transcription des opérations ci-dessus en TSL (Manuel annexe B page 255)

Bon courage à tous.

Le contenu du présent document composé sur Macintosh II et PageMaker donne l'information la plus complète et la plus exacte possible au moment de la publication, mais n'est ni garanti quant à la complétude, ni quant à l'exactitude. **Aware, Omnis, Quartz, Xchange, Archive, Abacus, Quill, Easel, PC4, Organiseur II, PageMaker, Apple II, IIGS, Macintosh, ImageWriter, LaserWriter, IBM** sont des marques déposées.
Création et réalisation maquette : SYLVIE LUCET

Dernières versions

MC

Système 1.29
Mobile TR 1.08
Mclink 1.29

ORGANISEUR

CM 3.3
XP 3.6
POS200 3.6
POS250 3.6
POS350 3.6
POS296 3.6

LZ 4.5
LZ64 4.5

Dévelop. 3.01
Diary link 1.0
CL PC 2.1
CL MAC 2.1

Log. 2 lignes
Tableur 2.0
Topfinance 2.2
Filepak 2.0
Barpak 0.9
Formulator 1.0
Travel Pak 1.5
Maths Pak 1.0

Log. 4 lignes
Tableur 2.0
Topfinance 1.3
Filepak 1.4
Prakpak 1.0

PC4

PC4 1.1

XCHANGE

Xchange 1.37